

UNIFIED MODEL DOCUMENTATION PAPER NO 7

**FILENAMING CONVENTIONS FOR
THE UNIFIED MODEL SYSTEM**

by

T.C.JOHN

VERSION NO. 4, DATED 16/06/92

[This describes the filenaming convention in use from UM release 2.7 onwards]

Climate Research
Meteorological Office
London Road
BRACKNELL
Berkshire
RG12 2SY
United Kingdom

(c) Crown Copyright 1996

This document has not been published. Permission to quote from it must be obtained from the Head of Climate Prediction at the above address.

Modification Record		
Document version	Author	Description.....

Filenaming Conventions for the Unified Model System

A filenaming convention for the UM should satisfy the requirements of all its potential users. On the CRAY, the filename system needs to take account of directory path name as well as the file name itself. The operational requirement for output filenames which are reused daily or sub-daily may appear to be at odds with the climate model requirement for unique filenames, but there is some scope for a unified approach to file naming, as outlined below.

Consideration of the inputs and outputs of the UM core system (ie. the standard model code and scripts) indicates that they can be classified into the following broad classes, as illustrated in Figure 1.

Input class 1

Simple input files not related to a specific model. Examples of this class are orography, land-sea masks, raw observations files and climatological forcing data.

Input class 2

Model-specific input files supplied by the user. These are likely to be mainly control files, and examples include the history file, diagnostic control file and UPDATES. Mostly, the names for these will be generated by the User Interface using the defined convention, with the exception of UPDATES.

Output class 1

Output files with predetermined names (ie. names set by the user). Operationally, all output files except restart dumps and some interface field files will fall into this class. Names of such files are essentially a free choice, but a naming convention should be observed wherever possible.

Output class 2

Output files with names determined by the model at run-time. Many output files (eg. mean dumps, PPfield files) will be of this type in climate models, since the run may be very long and it would be impractical for the user to specify possibly hundreds of different filenames. The experiment ID and validity time should be incorporated into the name of such files.

I/O class 1

Input/output files shared between two models (output from one, input to another) which are independent, in the sense of not being part of the same network. Examples include ocean surface forcing files derived from atmosphere-only climate runs, and starting dumps for climate experiments taken from a prior climate experiment.

I/O class 2

Input/output files shared between two models within a network ("communication files"). A strong naming convention is crucial for this type of file if flexible networking of models is to be supported in future. An example is boundary forcing for a regional model produced from a global model.

See figure 1. It should be clear into which class or classes a particular file fits. The filenaming convention should reflect the I/O class of the file. The following is a proposed definition of a suitable filenaming system for the UM, using the Backus-Naur Form (BNF) metalanguage for shorthand. This should be self-explanatory. A 14-character limit is observed (except for possible workstation extensions) for conformance with standard UNIX. The proposals relate mainly to filenames generated internally within the UM itself. Some files are prenamed by the user or User Interface; model-specific ones should observe a loose naming convention based on the experiment and job id, but the UM clearly cannot enforce naming conventions for user-defined filenames.

Proposed UM filename definition (CRAY/HDS/Workstation files)

These revised proposals cover the following extensions and revisions:

- a. Explicit submodel indicator (ie. atmos, ocean, wave, etc.)
- b. Redefinition of mean period indicator letters
- c. Extension to non-contiguous multi-year means
- d. Extension to allow multiple 'daily' PP output streams
- e. Extension to suggest workstation filenames
- f. Extension to allow 'friendly' monthly and seasonal mean period names
- g. Use of different separator letters to denote timestamp convention

Basic definitions :-

```
<letter> ::= a|b|..|z          <numeric> ::= 0|1|2|3|4|5|6|7|8|9
<alphabet> ::= <letter>|<numeric>
<month> ::= 1|2|3|..|9|a|b|c    <day> ::= 1|2|3|..|9|a|b|..|v
<hour> ::= 0|1|2|..|9|a|b|..|n
<month_3char> ::= jan|feb|mar|apr|may|jun|jul|aug|sep|oct|nov|dec
<month_2char> ::= jn|fb|mr|ar|my|jn|jl|ag|sp|ot|nv|dc
<season_3char> ::= djf|jfm|fma|mam|amj|mjj|jja|jas|aso|son|ond|ndj
<season_2char> ::= df|jm|fa|mm|aj|mj|ja|js|ao|sn|od|nj
<year_base_10> ::= <numeric>          (0-9 years)
<year_base_360> ::= <alphabet><numeric> (Up to 360 years)
<year_base_3600> ::= <alphabet><numeric><numeric> (Up to 3600 years)
<day_no> ::= <numeric><numeric><numeric>          (001-366)
<hours_since_data_time> ::= <numeric><numeric><numeric>
<timesteps_since_basis_time> ::= <numeric><numeric><numeric>
```

File name definitions

```
<CRAY_fullname> ::= <CRAY_directory>/<CRAY_filename>
<HDS_fullname>  ::= <HDS_pathname>.<HDS_filename>
<WS_fullname>   ::= <WS_directory>/<CRAY_filename>{<extension>}

<CRAY_directory> ::= <$DATAM>|<$DATAW>|<$HOME>/<expt_id>{/<subdirectory>}
<HDS_pathname>   ::= <HLQ>.<PL>
<WS_directory>  ::= <$TRANSDIR>/<expt_id>

<CRAY_filename> ::= <expt_id>{<job_id>}<submodel_id><separator><filename> |
                  <user_specified_filename> | <UI_specified_filename>
<HDS_filename>  ::= <expt_id>{<job_id>}<submodel_id>{<separator>}.<filename>
<user_specified_filename> ::= Up to 14 characters specified by user
<UI_specified_filename> ::= <expt_id>{<job_id>}.<simple_name>
<simple_name> ::= Up to 8 characters giving 'meaningful' name of file
<expt_id> ::= 4 char. experiment name or operational alias including prefix
<job_id>  ::= 1 character job identifier (omitted if operational)
<submodel_id> ::= a (Atmosphere) | o (Ocean) | w (Wave) | ...
<separator> ::= _ (Indicates relative_time timestamp format) |
               . (Indicates standard_absolute_time timestamp or partial
               sum) |
               - (Indicates short_absolute_time timestamp) |
               @ (Indicates long_absolute_time timestamp)
```

NB: IF <separator> = . or _ it is omitted from <HDS_filename>

```
<filename> ::= <file_type_1><file_type_2><timestamp>
```

```

<file_type_1>::= d (Dump) | s (partial Sum dump) |
                p (daily or standard mean PP file) |
                m (Multi-year Mean or other time-processed pp file) |
                b (Boundary forcing file) | n (aNCillary forcing file) |
                r (Reformatted raw observations file) |
                q (Quality-controlled observations file) |
                a (AC observations file) |

```

```

<file_type_2>::=CASE <file_type_1> IN

```

```

    d : a (instAntaneous forecast dump) |
        z (instantaneous before T=Zero in assimilation runs) |
        t (Ten-day mean) | m (Monthly) | s (Seasonal) | y (1-Year
          mean) |
        1 (period 1 mean if none of above) | 2 | 3 | 4 (similarly)

    s : 1 | 2 | 3 | 4 (Period 1,2,3,4)

    p : a|b|c|d|e|f|g|h|i|j (instantaneous stream A,B,C,D,E,F,G,H,I,J) |
        p (Pentad mean 5-day) | w (Week mean 7-day) | t (Ten-day mean) |
        r (fortnight mean 14-day) |
        m (Monthly) | s (Seasonal) | y (1-Year mean) |
        v (5-year mean) | x (10-year mean) | 1 (50-year mean) |
        u (100-year mean) | z (1000-year mean) |
        1 (period 1 mean if none of above) | 2 | 3 | 4 (similarly) |
        5 (user-defined period if none of the above)

    m : a|b|c|d|e|f|g|h|i|j
        (multi-annual mean of instantaneous stream A etc) |
        t (multi-annual Ten-day mean) | m (multi-annual Monthly) |
        s (multi-annual Seasonal) |
        i-annual period 1 if none of above) | 2 | 3 | 4 (similar)

    r,q,a : s (Surface) | r (Radiosonde) | b (satoB) | t (saTem) |
           a (Airep) | u (Uars) | c (Combined)
           [ but may need extending for ocean? ]

    b : 1|2|3|4|5|6|7|8|9 (days) | t (Ten-days) | m (1-Month) |
        s (1-Season) | y (1-Year) | b (other)

    n : n [ future extensions possible perhaps? ]

```

ESAC

```

<timestamp>::=<relative_time>|<short_absolute_time>|
             <standard_absolute_time>|<long_absolute_time>|
             <partial_sum_toggle>

```

```

<partial_sum_toggle>::= a|b

```

```

<relative_time>::=<hours_since_data_time>|<timesteps_since_basis_time>

```

```

<short_absolute_time>::=          <year_base_10><day_no><hour>
<standard_absolute_time>::=      <year_base_360><year_date_3>
<long_absolute_time>::=          <year_base_3600><year_date_2>

```

```

<year_date_3>::=CASE <file_type_1> IN
    d,p,m: CASE <file_type_2> IN
        m: <month_3char>|<month><day><hour>
        s: <season_3char>|<month><day><hour>
        *: <month><day><hour>
    ESAC
    *; <month><day><hour>
ESAC

```

```

<year_date_2>::=CASE <file_type_1> IN

```

```

        d,p,m: CASE <file_type_2> IN
            m: <month_2char>|<month><day>
            s: <season_2char>|<month><day>
        *: <month><day>
    ESAC
    *: <month><day>
    ESAC
<extension>::= CASE <file_type_1> IN
    p,m: .pp (If >14 chars allowed)
    * : <null>
    ESAC

```

NB: <extension> only allowed on workstations supporting long names.

Summary (generic filename excluding extension)

EXPT ID	JOB ID	SUBMODEL ID	SEPARATOR	FILE TYPE	FILE TYPE	FILE	FILE	TIMESTAMP
				1	2			
qwexpt		a	-	d	{a tmsy 1234}			<relative_time>
expt j		o	.	p	{abcdefghij tmsy 12345}			<stndrd_absolute_time>
		w	-	m	{abcdefghij tmsy 1234}			<short_absolute_time>
			@	o	{srbtauc}			<long_absolute_time>
				b	{123456789 tmsy b}			
				n	n			
				s	{1234}			{ab}

Examples (CRAY/WS filenames)

-
1. exptja.psj2djf Atmosphere seasonal mean PP file ending 0Z 1 Mar 1992
 2. exptjo.pmj2feb Ocean monthly mean PP file ending 0Z 1 Mar 1992
 3. exptja.msj6jja Atmosphere multi-year seasonal mean ending 0Z 1 Sep 1996
 4. exptja.pxl1c10 Atmosphere 10-year mean ending 0Z 1 Dec 2001
 5. exptja@pxk01c1 Atmosphere 10-year mean ending 1 Dec 2001 (long form)
 6. exptjw-pc21256 Wave model instantaneous PP stream C ending at 06Z Day number 125
1992
 7. exptja.s1a Atmosphere period 1 partial sum dump a
 8. exptjo.s4b cean period 4 partial sum dump b
 9. exptja.nnj2310 Atmosphere ancillary forcing file ending 0Z 1 Mar 1992
 10. exptja.bmj2feb Atmosphere boundary forcing file ending 0Z 1 Mar 1992
covering previous month (february)
 11. qwumgla_da036 Operational global forecast atmosphere dump at T+36
 12. qwoplaa_da003 Operational limited-area forecast dump at T+03
 13. qwopgla_dz003 Operational global forecast atmosphere dump at T-03
(during assimilation before analysis time)

Notes

- 1 A given file class will conform to the appropriate definitions, as outlined above.
- 2 The subchoice within <timestamp> will be determined globally for a given job depending on the user's application. Climate models will generally use <standard_absolute_time>, whereas operational models, which require output filenames to be the same every day, will use <relative_time> definitions.
- 3 <expt_id> is defined to be the operational experiment alias plus prefix in operational runs. This allows multiple operational runs to use the same filename aliases (eg. all global runs of the forecast to use the same <expt_alias>, and therefore share the same dump filenames). All other runs will generate unique names based on the 4-character expt_id set by the UI.
- 4 User Interface naming conventions for control files on the front-end are documented separately.
- 5 Directory structures, and therefore <CRAY_directory>, conforms loosely to a convention as outlined in the document "The Operational Suite on the CRAY" (Version 2, dated 22/1/90). The convention is emphasized in the filename itself in the <operational_file_prefix> for operational files, but not for non-operational ones. The only crucial aspect of this as far as general users are concerned is that output files with model-generated names are all written to the same directory (referenced via pointer DATAM), which is the working directory during model execution.
- 6 <job_id> is used only if it is not an operational run.

